USGS
science for a changing world

Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

A MODULAR THREE-DIMENSIONAL FINITE-DIFFERENCE GROUND-WATER FLOW MODEL

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological Survey Open-File Report 83–875

Book 6

MODELING TECHNIQUES

Module Documentation for the General-Head Boundary Package

The General-Head Boundary Package (GHB1) consists of four modules, all of which are called by the MAIN program. The modules are:

GHB1AL — Allocates space for an array that contains the general-head boundary list (BNDS).

GHB1RP — Reads location, boundary head, and boundary conductance (C_m) of each cell containing general-head boundary m.

GHB1FM — Adds the terms $-C_m$ and $-C_mHB_m$ to the accumulators $HCOF_{i,j,k}$ and $RHS_{i,j,k}$, respectively.

GHB1BD — Calculates the rates and accumulated volume of flow to and from general-head boundaries.

Narrative for Module GHB1AL

This module allocates space in the X array to store the list of general-head boundaries (GHB).

1. Print a message identifying the package and initialize NBOUND (number of general-head boundaries).

2. Read and print MXBND (the maximum number of general-head boundaries) and IGHBCB (the unit number for saving cell-by-cell flow terms or a flag indicating that cell-by-cell flow terms should be printed).

3. Set LCBNDS, which will point to the first element in the boundary list (BNDS), equal to ISUM which is currently pointing to the first unallocated element in the X array.

4. Calculate the amount of space needed for the boundary list (five values for each boundary--row, column, layer, head, and conductance) and add it to ISUM so that it continues to point to the first unallocated element in X.

5. Print the number of elements in the X array used by the GHB Package.

6. RETURN.

Flow Chart for Module GHB1AL

NBOUND is the number of general-head boundaries being simulated at any given time.

MXBND is the maximum number of general-head boundaries simulated.

IGHBCB is a flag and a unit number.

If IGHBCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set.

If IGHBCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IGHBCB < 0, the boundary leakage for each cell will be printed whenever ICBCFL is set.

LCBNDS is the location in the X array of the list of general-head boundaries data (BNDS).

BNDS is a table containing data for general-head boundaries.



```
      SUBROUTINE GHB1AL(ISUM,LENX,LCBNDS,NBOUND,MXBND,IN,IOUT,
     1                  IGHBCB)
C
C-----VERSION 1610 12MAY1987 GHB1AL
C     ******************************************************************
C     ALLOCATE ARRAY STORAGE FOR HEAD-DEPENDENT BOUNDARIES
C     ******************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------------
C     ------------------------------------------------------------------
C
C1------IDENTIFY PACKAGE AND INITIALIZE # OF GENERAL HEAD BOUNDS
      WRITE(IOUT,1)IN
    1 FORMAT(1H0,'GHB1 -- GHB PACKAGE, VERSION 1, 9/1/87',
     1' INPUT READ FROM UNIT',I3)
      NBOUND=0
C
C2------READ AND PRINT MXBND AND IGHBCB (MAX # OF BOUNDS AND UNIT
C2------FOR CELL-BY-CELL FLOW TERMS FOR GHB)
      READ(IN,2) MXBND,IGHBCB
    2 FORMAT(2I10)
      WRITE(IOUT,3) MXBND
    3 FORMAT(1H ,'MAXIMUM OF',I5,' HEAD-DEPENDENT BOUNDARY NODES')
      IF(IGHBCB.GT.0) WRITE(IOUT,9) IGHBCB
    9 FORMAT(1X,'CELL-BY-CELL FLOW WILL BE RECORDED ON UNIT',I3)
      IF(IGHBCB.LT.0) WRITE(IOUT,8)
    8 FORMAT(1X,'CELL-BY-CELL FLOW WILL BE PRINTED WHEN ICBCFL NOT 0')
C
C3------SET LCBNDS EQUAL TO ADDRESS OF FIRST UNUSED SPACE IN X.
      LCBNDS=ISUM
C
C4------CALCULATE AMOUNT OF SPACE USED BY THE GENERAL HEAD LIST.
      ISP=5*MXBND
      ISUM=ISUM+ISP
C
C5------PRINT AMOUNT OF SPACE USED BY THE GHB PACKAGE
      WRITE(IOUT,4) ISP
    4 FORMAT(1X,I8,' ELEMENTS IN X ARRAY ARE USED FOR HEAD',
     1      '-DEPENDENT BOUNDARIES')
      ISUM1=ISUM-1
      WRITE(IOUT,5) ISUM1,LENX
    5 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED OUT OF',I8)
      IF(ISUM1.GT.LENX) WRITE(IOUT,6)
    6 FORMAT(1X,'   ***X ARRAY MUST BE DIMENSIONED LARGER***')
C
C6------RETURN
      RETURN
      END
```

List of Variables for Module GHB1AL

Variable	Range	Definition
IGHBCB	Package	Flag and a unit number. > 0, unit number on which the cell-by-cell flow terms will be recorded whenever ICBCFL is set. = 0, cell-by-cell flow terms will not be printed or recorded. < 0, boundary leakage for each cell will be printed whenever IGHBFL is set.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISP	Module	Number of words in the X array allocated by this module.
ISUM	Global	Index number of the lowest element in the X array which has not yet been allocated. When space is allocated for an array, the size of the array is added to ISUM.
ISUM1	Module	ISUM-1.
LCBNDS	Package	Location in the X array of the first element of array BNDS.
LENX	Global	Length of the X array in words. This should always be equal to the dimension of X specified in the MAIN program.
MXBND	Package	Maximum number of head boundaries active at any one time.
NBOUND	Package	Number of head boundaries active during the current stress period.

Narrative for Module GHB1RP

This module reads data to build the general-head boundary list.

1. Read ITMP. ITMP is the number of general-head boundaries or a flag indicating that data from the previous stress period should be reused.

2. Test ITMP. If ITMP is less than zero, the general-head boundary data read for the last stress period will be reused. Print a message to that effect and RETURN.

3. If ITMP is greater than or equal to zero, it is the number of general-head boundaries for this stress period. Set the number of general-head boundaries (NBOUND) in the current stress period equal to ITMP.

4. Compare the number of general-head boundaries (NBOUND) in the current stress period to the number specified as the maximum for the simulation (MXBND). If NBOUND is greater than MXBND, STOP.

5. Print the number of general-head boundaries in the current stress period (NBOUND).

6. See if there are any general-head boundaries. If there are none in the current stress period (NBOUND = 0), bypass further boundary processing (SKIP STEP 7).

7. Read and print the layer, row, column, head, and conductance for each general-head boundary.

8. RETURN.

Flow Chart for Module GHB1RP

ITMP is both a flag and a counter. If it is greater than or equal to zero, it is the number of general-head boundaries to be simulated during the stress period. If it is less than zero, it indicates that the boundaries simulated in the last stress period should be simulated in the current stress period.

MXBND is the maximum number of general-head boundaries to be simulated.



```
      SUBROUTINE GHB1RP(BNDS,NBOUND,MXBND,IN,IOUT)
C
C
C-----VERSION 1651 02FEB1983 GHB1RP
C     ******************************************************************
C     READ DATA FOR GHB
C     ******************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------------
      DIMENSION BNDS(5,MXBND)
C     ------------------------------------------------------------------
C
C1------READ ITMP(# OF GENERAL HEAD BOUNDS OR FLAG TO REUSE DATA.)
      READ(IN,8) ITMP
    8 FORMAT(I10)
C
C2------TEST ITMP
      IF(ITMP.GE.0) GO TO 50
C
C2A-----IF ITMP<0 THEN REUSE DATA FROM LAST STRESS PERIOD
      WRITE(IOUT,7)
    7 FORMAT(1HO,'REUSING HEAD-DEPENDENT BOUNDS FROM LAST STRESS',
     1      ' PERIOD')
      GO TO 260
C
C3------IF ITMP=>0 THEN IT IS THE # OF GENERAL HEAD BOUNDS.
   50 NBOUND=ITMP
C
C4------IF MAX NUMBER OF BOUNDS IS EXCEEDED THEN STOP
      IF(NBOUND.LE.MXBND) GO TO 100
      WRITE(IOUT,99) NBOUND,MXBND
   99 FORMAT(1HO,'NBOUND(',I4,') IS GREATER THAN MXBND(',I4,')')
C
C4A-----ABNORMAL STOP
      STOP
C
C5------PRINT # OF GENERAL HEAD BOUNDS THIS STRESS PERIOD
  100 WRITE(IOUT,1) NBOUND
    1 FORMAT(1HO,//1X,I5,' HEAD-DEPENDENT BOUNDARY NODES')
C
C6------IF THERE ARE NO GENERAL HEAD BOUNDS THEN RETURN.
      IF(NBOUND.EQ.0) GO TO 260
C
C7------READ & PRINT DATA FOR EACH GENERAL HEAD BOUNDARY.
      WRITE(IOUT,3)
    3 FORMAT(1HO,15X,'LAYER',5X,'ROW',5X
     1,'COL   ELEVATION   CONDUCTANCE   BOUND NO.'/1X,15X,60('-'))
      DO 250 II=1,NBOUND
      READ (IN,4) K,I,J,BNDS(4,II),BNDS(5,II)
    4 FORMAT(3I10,2F10.0)
      WRITE (IOUT,5) K,I,J,BNDS(4,II),BNDS(5,II),II
    5 FORMAT(1X,15X,I4,I9,I8,G13.4,G14.4,I8)
      BNDS(1,II)=K
      BNDS(2,II)=I
      BNDS(3,II)=J
  250 CONTINUE
C
C8------RETURN
  260 RETURN
      END
```

List of Variables for Module GHB1RP

Variable	Range	Definition
BNDS	Package	DIMENSION (5,MXBND), Layer, row, column, head and conductance from boundary for each general-head boundary.
I	Module	Row number.
II	Module	Index for general-head boundaries.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ITMP	Module	Flag or number of boundaries. $\geq$ 0, number of bounds active during the current stress period. < 0, same bounds active during the last stress period will be active during the current stress period.
J	Module	Column number.
K	Module	Layer number.
MXBND	Package	Maximum number of head boundaries active at any one time.
NBOUND	Package	Number of head boundaries active during the current stress period.

Narrative for Module GHB1FM

This module adds terms representing riverhead boundaries to the accumulators HCOF and RHS.

1. If NBOUND is less than or equal to zero in the current stress period, there are no general-head boundaries. RETURN.

2. For each boundary in the BNDS list, DO STEPS 3-6.

3. Determine the column (IC), row (IR), and layer (IL).

4. If the cell is external (IBOUND(IC, IR, IL) $\leq$ 0), bypass processing on this boundary and go on to the next one.

5. If the cell is internal, get the boundary data (head and conductance).

6. Add the -C*HB term (C is the conductance and HB is the boundary head) to the accumulator RHS and the term -C to the accumulator HCOF.

7. RETURN.

Flow Chart for Module GHB1FM

RHS is an accumulator in which the right hand side of the equation is formulated.

HCOF is an accumulator in which the coefficient of head in the cell is formulated.



```
      SUBROUTINE GHB1FM(NBOUND,MXBND,BNDS,HCOF,RHS,IBOUND,
     1                  NCOL,NROW,NLAY)
C
C-----VERSION 1037 10APR1985 GHB1FM
C     ******************************************************************
C     ADD GHB TERMS TO RHS AND HCOF
C     ******************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------------
      DIMENSION BNDS(5,MXBND),HCOF(NCOL,NROW,NLAY),
     1          RHS(NCOL,NROW,NLAY),IBOUND(NCOL,NROW,NLAY)
C     ------------------------------------------------------------------
C
C1------IF NBOUND<=0 THEN THERE ARE NO GENERAL HEAD BOUNDS. RETURN.
      IF(NBOUND.LE.0) RETURN
C
C2------PROCESS EACH ENTRY IN THE GENERAL HEAD BOUND LIST (BNDS)
      DO 100 L=1,NBOUND
C
C3------GET COLUMN, ROW AND LAYER OF CELL CONTAINING BOUNDARY
      IL=BNDS(1,L)
      IR=BNDS(2,L)
      IC=BNDS(3,L)
C
C4------IF THE CELL IS EXTERNAL THEN SKIP IT.
      IF(IBOUND(IC,IR,IL).LE.0) GO TO 100
C
C5------SINCE THE CELL IS INTERNAL GET THE BOUNDARY DATA.
      HB=BNDS(4,L)
      C=BNDS(5,L)
C
C6------ADD TERMS TO RHS AND HCOF
      HCOF(IC,IR,IL)=HCOF(IC,IR,IL)-C
      RHS(IC,IR,IL)=RHS(IC,IR,IL)-C*HB
  100 CONTINUE
C
C7------RETURN
      RETURN
      END
```

List of Variables for Module GHB1FM

Variable	Range	Definition
BNDS	Package	DIMENSION (5,MXBND), Layer, row, column, head and conductance from boundary for each general-head boundary.
C	Module	Conductance from the external boundary.
HB	Module	Head on boundary.
HCOF	Global	DIMENSION (NCOL,NROW,NLAY), Coefficient of head in the cell (J,I,K) in the finite-difference equation.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
L	Module	Index for boundaries.
MXBND	Package	Maximum number of head boundaries active at any one time.
NBOUND	Package	Number of head boundaries active during the current stress period.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
RHS	Global	DIMENSION (NCOL,NROW,NLAY), Right hand side of the finite-difference equation. RHS is an accumulation of terms from several different packages.

Narrative for Module GHB1BD

This module calculates rates and volumes transferred between the aquifer and general-head boundaries.

1. Initialize the cell-by-cell flow-term flag (IBD) and the rate accumulator (RATOUT).

2. If there are no general-head boundaries (NBOUND = 0), skip down to step 13 and put zeros into the budget terms for general-head boundaries.

3. Test to see if cell-by-cell flow terms are to be saved on disk. They will not be saved if either of the following conditions hold: (1) this is not the proper time step (ICBCFL = 0) or (2) cell-by-cell flow terms are not needed for general-head boundaries during this simulation (IGHBCB $\leq$ 0). If cell-by-cell flow terms will be saved for this package, clear the buffer in which they will be accumulated (BUFF) and set the cell-by-cell flow-term flag (IBD).

4. For each general-head boundary, DO STEPS 5-13 accumulating flows from or into the general-head boundary.

5. Determine the row, column, and layer of the cell containing the general-head boundary.

6. If the cell is external (IBOUND(I,J,K) $\leq$ 0), bypass further processing of this boundary.

7. Get the boundary parameters from the boundary list (BNDS).

8. Set RATE equal to the boundary conductance times the boundary head minus the head in the cell (RATE = C*(HB - HHNEW)).

9. If cell-by-cell flow terms are to be printed (IGHBCB < 0 and ICBCFL ≠ 0), print RATE.

10. If budget terms for individual cells are to be saved, add the RATE to the buffer (BUFF).

11. Check to see whether flow is into or out of the aquifer.

12. If RATE is negative, add it to RATOUT.

13. If RATE is positive, add it to RATIN.

14. See if cell-by-cell flow terms for individual cells are to be saved (IBD = 1). If they are, call module UBUDSV to record the buffer (BUFF) onto disk.

15. Move RATIN and RATOUT into the VBVL array for printing by BAS1OT. Add RATIN and RATOUT multiplied by the time-step length to the volume accumulators in VBVL for printing by BAS1OT. Move the general-head boundary budget-term labels to VBNM for printing by BAS1OT.

16. Increment the budget-term counter (MSUM). See the section in the Basic Package for a detailed explanation of VBVL, VBNM, and MSUM.

17. RETURN.

Flow Chart for Module GHB1BD

IBD is a flag which, if set, causes cell-by-cell flow terms for general-head boundary to be recorded.

EXTERNAL: a cell is said to be external if it is either no flow or constant head (i.e., an equation is not formulated for the cell).

RATE is the leakage rate into the aquifer from the boundary in a cell.

BUFFER is an array in which values are stored as they are being gathered for printing or recording.

RATOUT is an accumulator to which all flows out of the aquifer are added.

RATIN is an accumulator to which all flows into the aquifer are added.

C is the conductance between the boundary and the cell.

HB is the boundary head.

HHNEW is the head in the cell.

IGHBCB is a flag and a unit number.

If IGHBCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set.

If IGHBCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IGHBCB < 0, boundary leakage for each cell will be printed whenever ICBCFL is set.

ICBCFL is a flag.

If ICBCFL ≠ 0, cell-by-cell flow terms will be either recorded or printed depending on IGHBCB for the current time step.



```
      SUBROUTINE GHB1BD(NBOUND,MXBND,VBNM,VBVL,MSUM,BNDS,DELT,HNEW,
     1    NCOL,NROW,NLAY,IBOUND,KSTP,KPER,IGHBCB,ICBCFL,BUFF,IOUT)
C
C-----VERSION 1612 12MAY1987 GHB1BD
C     ******************************************************************
C     CALCULATE VOLUMETRIC BUDGET FOR GHB
C     ******************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------------
      CHARACTER*4 VBNM,TEXT
      DOUBLE PRECISION HNEW
      DIMENSION VBNM(4,MSUM),VBVL(4,MSUM),BNDS(5,MXBND),
     1          HNEW(NCOL,NROW,NLAY),IBOUND(NCOL,NROW,NLAY),
     2          BUFF(NCOL,NROW,NLAY)
      DIMENSION TEXT(4)
      DATA TEXT(1),TEXT(2),TEXT(3),TEXT(4) /' HEA','D DE','P BO','UNDS'/
C     ------------------------------------------------------------------
C
C1------INITIALIZE CELL-BY-CELL FLOW TERM FLAG (IBD) AND
C1------ACCUMULATORS (RATIN AND RATOUT)
      IBD=0
      RATOUT=0.
      RATIN=0.
C
C2------IF NO BOUNDARIES THEN KEEP ZEROES IN ACCUMULATORS.
      IF(NBOUND.EQ.0) GO TO 200
C
C3------TEST TO SEE IF CELL-BY-CELL FLOW TERMS ARE NEEDED.
      IF(ICBCFL.EQ.0 .OR. IGHBCB.LE.0) GO TO 10
C
C3A-----SINCE CELL-BY-CELL FLOW TERMS ARE NEEDED CLEAR BUFFER & SET
C3A-----THE FLAG IBD.
      IBD=1
      DO 5 IL=1,NLAY
      DO 5 IR=1,NROW
      DO 5 IC=1,NCOL
      BUFF(IC,IR,IL)=0.
    5 CONTINUE
C
C4------FOR EACH GENERAL HEAD BOUND ACCUMULATE FLOW INTO AQUIFER
   10 DO 100 L=1,NBOUND
C
C5------GET LAYER, ROW AND COLUMN OF EACH GENERAL HEAD BOUNDARY.
      IL=BNDS(1,L)
      IR=BNDS(2,L)
      IC=BNDS(3,L)
C
C6------IF CELL IS EXTERNAL THEN IGNORE IT.
      IF(IBOUND(IC,IR,IL).LE.0) GO TO 100
```

```
C
C7------GET PARAMETERS FROM BOUNDARY LIST.
      HHNEW=HNEW(IC,IR,IL)
      HB=BNDS(4,L)
      C=BNDS(5,L)
C
C8------CALCULATE THE FOW RATE INTO THE CELL
      RATE=C*(HB-HHNEW)
C
C9------PRINT THE INDIVIDUAL RATES IF REQUESTED(IGHBCB<0).
      IF(IGHBCB.LT.0.AND.ICBCFL.NE.0) WRITE(IOUT,900) (TEXT(N),N=1,4),
     1    KPER,KSTP,L,IL,IR,IC,RATE
  900 FORMAT(1H0,4A4,'   PERIOD',I3,'   STEP',I3,'   BOUNDARY',I4,
     1    '   LAYER',I3,'   ROW',I4,'   COL',I4,'   RATE',G15.7)
C
C10-----IF CELL-BY-CELL TERMS ARE TO BE SAVED THEN PUT RATE IN BUFFER
      IF(IBD.EQ.1) BUFF(IC,IR,IL)=BUFF(IC,IR,IL)+RATE
C
C11-----SEE IF FLOW IS INTO AQUIFER OR OUT OF AQUIFER.
      IF(RATE)94,100,96
C
C12------FLOW IS OUT OF AQUIFER SUBTRACT RATE FROM RATOUT
   94 RATOUT=RATOUT-RATE
      GO TO 100
C
C13-----FLOW IS INTO AQIFER ADD RATE TO RATIN
   96 RATIN=RATIN+RATE
  100 CONTINUE
C
C14-----IF CELL-BY-CELL TERMS ARE TO BE SAVED THEN CALL
C14-----UTILITY MODULE UBUDSV
      IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,IGHBCB,BUFF,NCOL,NROW,
     1                         NLAY,IOUT)
C
C15-----MOVE RATES, VOLUMES AND LABELS INTO ARRAYS FOR PRINTING
  200 VBVL(3,MSUM)=RATIN
      VBVL(1,MSUM)=VBVL(1,MSUM)+RATIN*DELT
      VBVL(4,MSUM)=RATOUT
      VBVL(2,MSUM)=VBVL(2,MSUM)+RATOUT*DELT
      VBNM(1,MSUM)=TEXT(1)
      VBNM(2,MSUM)=TEXT(2)
      VBNM(3,MSUM)=TEXT(3)
      VBNM(4,MSUM)=TEXT(4)
C
C16-----INCREMENT THE BUDGET TERM COUNTER
      MSUM=MSUM+1
C
C17-----RETURN
      RETURN
      END
```

List of Variables for Module GHB1BD

Variable	Range	Definition
BNDS	Package	DIMENSION (5,MXBND), Layer, row, column, head and conductance from the boundary for each general-head boundary.
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
C	Module	Conductance from the external boundary.
DELT	Global	Length of the current time step.
HB	Module	Head on boundary.
HHNEW	Module	HNEW (J,I,K), Single precision.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
IBD	Package	Flag. = 0, cell-by-cell flow terms for this package will not be recorded. ≠ 0, cell-by-cell flow terms for this package will be recorded.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
IGHBCB	Package	Flag and a unit number. > 0, unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set. = 0, cell-by-cell flow terms will not be printed or recorded. < 0, boundary leakage for each cell will be printed whenever ICBCFL is set.
ICBCFL	Global	Flag. = 0, cell-by-cell flow terms will not be recorded or printed for the current time step. ≠ 0, cell-by-cell flow terms will be either printed or recorded (depending on IGHBCB) for the current time step.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
L	Module	Index for general-head boundaries.
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
MXBND	Package	Maximum number of head boundaries active at any one time.
NBOUND	Package	Number of head boundaries active during the current stress period.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.

List of Variables for Module GHB1BD (Continued)

Variable	Range	Definition
RATE	Module	Flow from a bound into a cell. (Reverse the sign to get flow into the bound.)
RATIN	Module	Accumulator for the total flow into the flow field out of the bounds.
RATOUT	Module	Accumulator for the total flow out of the flow field into the bounds.
TEXT	Module	Label to be printed or recorded with the array data.
VBNM	Global	DIMENSION (4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION (4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N), Rate for the current time step into the flow field. (2,N), Rate for the current time step out of the flow field. (3,N), Volume into the flow field during simulation. (4,N), Volume out of the flow field during simulation.